
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.3.2.10 which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2012-03_D12wk48’.
The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_3_2_10
Test Group:
8_3\RRC_Measurement_C2K.ttcn
ATS Version:
iwd-EUTRA-B2012-03_D12wk48
System Simulator used:
R&S CMW500 
UE used:
LG Electronics Optimus G
Verification Status:
PASS


4. Corrections required for test case 8.3.2.10
This section describes the TTCN changes required to make test case 8.3.2.10 run correctly with a LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2012-03_D12wk48’ release.

Change 1 – Correction to group specific templates in module ‘RRC_Measurement_C2K.ttcn’ 
	Template name
	cs_RRCConnectionReconfiguration_IratPerMeas_C2K()
cr_MeasurementReport_CDMA2000_2Entries()

	Reason for change
	1.    TTCN implementation of “ReportConfigInterRAT-PERIODICAL” IE in the send template “cs_RRCConnectionReconfiguration_IratPerMeas_C2K()” was not as per default IE definition in TS 36.508 Table 4.6.6-9.
2.    As per TS 36.523-1 Table 8.3.2.10.3.3-2 the MeasConfig sequence of the RRCConnectionReconfiguration message in step 1 contains only “pilot strength” as CDMA2000 measurement object. Therefore, measurement results for additional CDMA2000 measurement objects cannot be expected in measurement reports sent by the UE in later test steps 6, 7. 

	Summary of change
	1.    Corrected the “ReportConfigInterRAT-PERIODICAL” IE as per 36.508 requirements.
2.    Replaced some “?” with “omit” in receive template “cr_MeasResultNeighCells_measResultsCDMA2000_2Entries()” which is used in “cr_MeasurementReport_CDMA2000_2Entries()” at step 6, 7. 

	TTCN module
	8_3\RRC_Measurement_C2K.ttcn

	MCC160 Comment
	


Before change

…
  //8.3.2.8, 8.3.2.10 Step1
  template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_IratPerMeas_C2K(RRC_TransactionIdentifier p_RRC_TI,
                                                                                    ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq, //@sic CR R5-112614 sic@
                                                                                     AllowedMeasBandwidth p_EUTRA_MeasurementBandwidth, //@sic CR R5-112614 sic@
                                                                                   CDMA2000_Type p_CDMA2000_Type,
                                                                                   ARFCN_ValueCDMA2000 p_ARFCNCDMA2000,
                                                                                   BandclassCDMA2000 p_BandClassCDMA2000,
                                                                                   SearchWindowSize_Type p_SearchWindowSize,
                                                                                   MeasObjectId p_MeasObjectId,
                                                                                   MeasId p_ReportConfigId , //@sic R5-123656 sic@
                                                                                   template (value) CellsToAddModListCDMA2000 p_CellsToAddModList) :=
    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,
                                                      cs_MeasConfig(omit,
                                                                    cs_MeasObjectToAddModList_2Entries(tsc_IdMeasObject_f1, //@sic CR R5-112614 sic@
                                                                                                       cs_508_MeasObjectEUTRA_GENERIC( p_EUTRA_DL_CarrierFreq, p_EUTRA_MeasurementBandwidth ), //@sic CR R5-112614 sic@
                                                                                                       p_MeasObjectId,
                                                                                                     cs_MeasObject_measObjectCDMA2000(p_CDMA2000_Type,
                                                                                                                                      p_ARFCNCDMA2000,
                                                                                                                                      p_BandClassCDMA2000,
                                                                                                                                      p_SearchWindowSize,
                                                                                                                                      0,
                                                                                                                                      omit,
                                                                                                                                      p_CellsToAddModList,
                                                                                                                                      omit ) ),
                                                                    omit,
                                                                    cs_ReportConfigToAddModList_1Entry(p_ReportConfigId,
                                                                                                       cs_ReportConfig_reportConfigInterRAT_periodical(reportStrongestCells, 8, ms2048)),
                                                                    omit,
                                                                    cs_MeasIdToAddModList_1Entry( 1, p_MeasObjectId, p_ReportConfigId ),
                                                                    cs_QuantityConfig( omit, omit, omit, cs_QuantityConfigCDMA2000( pilotStrength ) ),
                                                                    cs_508_MeasGapConfig_GP2 // currently using this since the MeasGapConfig for CDMA2000 is not defined and the MeasGapConfig given in 36523 in test case prose cannot be found
                                                                    ) );
  //8.3.2.8, 8.3.2.10 Step10
  template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_MeasRemove( RRC_TransactionIdentifier p_RRC_TI ) :=
    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,
                                                      cs_MeasConfig(omit,
                                                                    omit,
                                                                    omit,
                                                                    omit,
                                                                    cs_MeasIdToRemoveList_1Entry( 1 ),
                                                                    omit ) );
  //------------------------------------------------
  //8.3.2.8, 8.3.2.10 Steps 6,7
  template UL_DCCH_Message cr_MeasurementReport_CDMA2000_2Entries( PhysCellIdCDMA2000 p_PhysCellId1,
                                                                   PhysCellIdCDMA2000 p_PhysCellId2 ) :=
    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultsCDMA2000_2Entries( false, p_PhysCellId1, p_PhysCellId2, omit, omit, ?, ?, ?, ? ) );
  //------------------------------------------------
  //measurementReport_r8.measResultNeighCells(CHOICE: MeasResultListCDMA2000)
  //cr_MeasResultNeighCells_measResultListCDMA2000_2Entries
  template MeasResults.measResultNeighCells cr_MeasResultNeighCells_measResultsCDMA2000_2Entries(template boolean p_PreRegistrationStatusHRPD,
                                                                                                 template PhysCellIdCDMA2000 p_PhysCellId15,
                                                                                                 template PhysCellIdCDMA2000 p_PhysCellId17,
                                                                                                 template MeasResultCDMA2000.cgi_Info p_Cgi_Info_cell15,
                                                                                                 template MeasResultCDMA2000.cgi_Info p_Cgi_Info_cell17,
                                                                                                 template CDMA2000_PilotPnPhase p_PilotPnPhase_cell15,
                                                                                                 template CDMA2000_PilotPnPhase p_PilotPnPhase_cell17,
                                                                                                 template CDMA2000_PilotStrength p_PilotStrength_cell15,
                                                                                                 template CDMA2000_PilotStrength p_PilotStrength_cell17 ) :=
  {
    measResultsCDMA2000 := {
      preRegistrationStatusHRPD := p_PreRegistrationStatusHRPD,
      measResultListCDMA2000 := {
        {
          physCellId := p_PhysCellId15,
          cgi_Info := p_Cgi_Info_cell15,
          measResult := {
            pilotPnPhase := p_PilotPnPhase_cell15,
            pilotStrength := p_PilotStrength_cell15
          }
        },
        {
          physCellId := p_PhysCellId17,
          cgi_Info := p_Cgi_Info_cell17,
          measResult := {
            pilotPnPhase := p_PilotPnPhase_cell17,
            pilotStrength := p_PilotStrength_cell17
          }
        }
      }
    }
  };  
…
After change

…
  //8.3.2.8, 8.3.2.10 Step1
  template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_IratPerMeas_C2K(RRC_TransactionIdentifier p_RRC_TI,
                                                                                    ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq, //@sic CR R5-112614 sic@
                                                                                     AllowedMeasBandwidth p_EUTRA_MeasurementBandwidth, //@sic CR R5-112614 sic@
                                                                                   CDMA2000_Type p_CDMA2000_Type,
                                                                                   ARFCN_ValueCDMA2000 p_ARFCNCDMA2000,
                                                                                   BandclassCDMA2000 p_BandClassCDMA2000,
                                                                                   SearchWindowSize_Type p_SearchWindowSize,
                                                                                   MeasObjectId p_MeasObjectId,
                                                                                   MeasId p_ReportConfigId , //@sic R5-123656 sic@
                                                                                   template (value) CellsToAddModListCDMA2000 p_CellsToAddModList) :=
    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,
                                                      cs_MeasConfig(omit,
                                                                    cs_MeasObjectToAddModList_2Entries(tsc_IdMeasObject_f1, //@sic CR R5-112614 sic@
                                                                                                       cs_508_MeasObjectEUTRA_GENERIC( p_EUTRA_DL_CarrierFreq, p_EUTRA_MeasurementBandwidth ), //@sic CR R5-112614 sic@
                                                                                                       p_MeasObjectId,
                                                                                                     cs_MeasObject_measObjectCDMA2000(p_CDMA2000_Type,
                                                                                                                                      p_ARFCNCDMA2000,
                                                                                                                                      p_BandClassCDMA2000,
                                                                                                                                      p_SearchWindowSize,
                                                                                                                                      0,
                                                                                                                                      omit,
                                                                                                                                      p_CellsToAddModList,
                                                                                                                                      omit ) ),
                                                                    omit,
                                                                    cs_ReportConfigToAddModList_1Entry(p_ReportConfigId,
                                                                                                       cs_ReportConfig_reportConfigInterRAT_periodical(reportStrongestCells, 1, ms1024, infinity_)),
                                                                    omit,
                                                                    cs_MeasIdToAddModList_1Entry( 1, p_MeasObjectId, p_ReportConfigId ),
                                                                    cs_QuantityConfig( omit, omit, omit, cs_QuantityConfigCDMA2000( pilotStrength ) ),
                                                                    cs_508_MeasGapConfig_GP2 // currently using this since the MeasGapConfig for CDMA2000 is not defined and the MeasGapConfig given in 36523 in test case prose cannot be found
                                                                    ) );
  //8.3.2.8, 8.3.2.10 Step10
  template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_MeasRemove( RRC_TransactionIdentifier p_RRC_TI ) :=
    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,
                                                      cs_MeasConfig(omit,
                                                                    omit,
                                                                    omit,
                                                                    omit,
                                                                    cs_MeasIdToRemoveList_1Entry( 1 ),
                                                                    omit ) );
  //------------------------------------------------
  //8.3.2.8, 8.3.2.10 Steps 6,7
  template UL_DCCH_Message cr_MeasurementReport_CDMA2000_2Entries( PhysCellIdCDMA2000 p_PhysCellId1,
                                                                   PhysCellIdCDMA2000 p_PhysCellId2 ) :=
    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultsCDMA2000_2Entries( false, p_PhysCellId1, p_PhysCellId2, omit, omit, omit, omit, ?, ? ) );
  //------------------------------------------------
  //measurementReport_r8.measResultNeighCells(CHOICE: MeasResultListCDMA2000)
  //cr_MeasResultNeighCells_measResultListCDMA2000_2Entries
  template MeasResults.measResultNeighCells cr_MeasResultNeighCells_measResultsCDMA2000_2Entries(template boolean p_PreRegistrationStatusHRPD,
                                                                                                 template PhysCellIdCDMA2000 p_PhysCellId15,
                                                                                                 template PhysCellIdCDMA2000 p_PhysCellId17,
                                                                                                 template MeasResultCDMA2000.cgi_Info p_Cgi_Info_cell15,
                                                                                                 template MeasResultCDMA2000.cgi_Info p_Cgi_Info_cell17,
                                                                                                 template CDMA2000_PilotPnPhase p_PilotPnPhase_cell15,
                                                                                                 template CDMA2000_PilotPnPhase p_PilotPnPhase_cell17,
                                                                                                 template CDMA2000_PilotStrength p_PilotStrength_cell15,
                                                                                                 template CDMA2000_PilotStrength p_PilotStrength_cell17 ) :=
  {
    measResultsCDMA2000 := {
      preRegistrationStatusHRPD := p_PreRegistrationStatusHRPD,
      measResultListCDMA2000 := {
        {
          physCellId := p_PhysCellId15,
          cgi_Info := p_Cgi_Info_cell15,
          measResult := {
            pilotPnPhase := p_PilotPnPhase_cell15,
            pilotStrength := p_PilotStrength_cell15
          }
        },
        {
          physCellId := p_PhysCellId17,
          cgi_Info := p_Cgi_Info_cell17,
          measResult := {
            pilotPnPhase := p_PilotPnPhase_cell17,
            pilotStrength := p_PilotStrength_cell17
          }
        }
      }
    }
};

…  
Change 2 – Corrections to function ‘f_TC_8_3_2_10_EUTRA()’
	Function name
	f_TC_8_3_2_10_EUTRA()

	Reason for change
	1.    Wrong index was used when configuring cell parameters for cell 20 just before start of the test body.
2.    In step 3-4 the “Timer 1” should have been stopped before timeout occurs. Otherwise, the TC would have failed unexpectedly in case of timeout.

3.   As per TS 36.523-1 Table 8.3.2.10.3.3-2, measurement reporting was only configured for one value only (“pilot strength”). Therefore, the second parameter in the receive template “cr_MeasurementReport_CDMA2000_1Entry()” in step 3-4 should have been set to “omit” instead of “?”. 
4.    In step 6-7 the “Timer 1” should have been stopped before timeout occurs. Otherwise, the TC would have failed unexpectedly in case of timeout.
5.   TS 36.523-1 Table 8.3.2.10.3.3-2 refers to the default definition of “ReportConfigInterRAT-PERIODICAL” IE in which “maxReportCells” is 1 acc. to TS 36.508 Tabel 4.6.6-9. Therefore, template “cr_MeasurementReport_CDMA2000_2Entries()” should not have been used in step 6-7 even though cell 19 and cell 20 are both “ON”.
6.   In step 9-10 the “Timer 1” should have been stopped before timeout occurs. Otherwise, the TC would have failed unexpectedly in case of timeout.

7.   As per TS 36.523-1 Table 8.3.2.10.3.3-2, measurement reporting was only configured for one value only (“pilot strength”). Therefore, the second parameter in the receive template “cr_MeasurementReport_CDMA2000_1Entry()” in step 9-10 should have been set to “omit” instead of “?”.

	Summary of change
	1.    Replaced index [1] with [0] in variable “v_CDMA2000SysInfo.Cdma2000”.
2.    Stopped Timer 1 in case of timeout in step 3-4.
3.    Replaced “?” with “omit” in “cr_MeasurementReport_CDMA2000_1Entry()” at step 3-4.
4.    Stopped Timer 1 in case of timeout in step 6-7.
5.   Template “cr_MeasurementReport_CDMA2000_1Entry()” used in step 6-7.
(This change requires a prose CR (see draft attachment in R5s120914.zip)
6.    Stopped Timer 1 in case of timeout in step 9-10.
7.    Replaced “?” with “omit” in “cr_MeasurementReport_CDMA2000_1Entry()” at step 9-10.

	TTCN module
	8_3\RRC_Measurement_C2K.ttcn

	MCC160 Comment
	


Before change

…
  //testcase 8.3.2.10:
  // Measurement configuration control and reporting / Inter-RAT measurements / Periodic reporting / Measurement of 1xRTT cells
  function f_TC_8_3_2_10_EUTRA () runs on EUTRA_PTC
  {
    var integer v_T0_T1_T2_RS_EPRE_Cell1 := -60;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var IRAT_CoOrd_SysInfo_Type v_CDMA2000SysInfo;
    var CDMA2000_Type  v_CDMA2000_Type_Cell19;
    var CDMA2000_Type  v_CDMA2000_Type_Cell20;
    var integer v_PNOffset_Cell19;
    var integer v_PNOffset_Cell20;
    var ARFCN_ValueCDMA2000 v_Arfcn_CDMA2000_Cell19;
    var ARFCN_ValueCDMA2000 v_Arfcn_CDMA2000_Cell20;
    var BandclassCDMA2000 v_BandClass_CDMA2000_Cell19;
    var BandclassCDMA2000 v_BandClass_CDMA2000_Cell20;
    var integer v_SearchWindowSize_Cell19 := 15;
    var float v_TimerValue_1;
    var float v_TimerValue_2;
    //@sic R5112614 sic@
    var Frequency_IE_Type v_Frequency_IE_Cell1;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell1 ;
    var default v_DefaultRef := null;
    timer t_Wait;
    timer t_CheckTimer_1;
    //Initialize testcase
    f_EUTRA_Init ( c6 );
    //@sic R5112614 sic@
    v_Frequency_IE_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell1);
    v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_IE_Cell1.DL_ChBandwidth, v_Frequency_IE_Cell1.UL_ChBandwidth);
    v_TimerValue_1 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, nonProtocolTimer, 30.0 );
    v_TimerValue_2 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, nonProtocolTimer, 5.0 );
    //Set EPRE for T0
    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, v_T0_T1_T2_RS_EPRE_Cell1 );
    //Create and configure Cell1
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    f_EUTRA_Preamble ( eutra_Cell1, STATE2_IDLEUPDATE );
    f_EUTRA_RbEst_Def (eutra_Cell1);
    // First Get Cell Parameters for CDMA2000 cell19
    v_CDMA2000SysInfo := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );
    v_CDMA2000_Type_Cell19 := f_ConvertFromBoolToCDMA2000_Type(v_CDMA2000SysInfo.Cdma2000[0].IsHRPD);
    v_PNOffset_Cell19 := v_CDMA2000SysInfo.Cdma2000[0].PNOffset;
    v_Arfcn_CDMA2000_Cell19 := v_CDMA2000SysInfo.Cdma2000[0].Arfcn;
    v_BandClass_CDMA2000_Cell19 := f_ConvertFromIntegerToBandClass(v_CDMA2000SysInfo.Cdma2000[0].BandClass);
    //v_SearchWindowSize_Cell19 := v_CDMA2000SysInfo.Cdma2000[0].SearchWindowSize;
    // Next Get Cell Parameters for CDMA2000 cell20
    v_CDMA2000SysInfo := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );
    v_CDMA2000_Type_Cell20 := f_ConvertFromBoolToCDMA2000_Type(v_CDMA2000SysInfo.Cdma2000[1].IsHRPD);
    v_PNOffset_Cell20 := v_CDMA2000SysInfo.Cdma2000[1].PNOffset;
    v_Arfcn_CDMA2000_Cell20 := v_CDMA2000SysInfo.Cdma2000[1].Arfcn;
    v_BandClass_CDMA2000_Cell20 := f_ConvertFromIntegerToBandClass(v_CDMA2000SysInfo.Cdma2000[1].BandClass);
    //v_SearchWindowSize_Cell20 := v_CDMA2000SysInfo.Cdma2000[1].SearchWindowSize;
    f_EUTRA_TestBody_Set ( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 1-2" siclog@
    //The SS transmits an RRCConnectionReconfiguration message including measConfig to setup measurements and periodical reporting for HRPD cells.
    //The UE transmits an RRCConnectionReconfigurationComplete message to confirm measurement on Cell 19.
    //@sic R5s110743 sic@
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig( eutra_Cell1, cs_RRCConnectionReconfiguration_IratPerMeas_C2K ( v_RRC_TI,
                                                                                       v_Frequency_IE_Cell1.UL_DL_Earfcn.dl_CarrierFreq, //@sic R5112614 sic@
                                                                                       v_ChBandDependency_Cell1.AllowedMeasBandwidth, //@sic R5112614 sic@
                                                                                       v_CDMA2000_Type_Cell19,
                                                                                       v_Arfcn_CDMA2000_Cell19,
                                                                                       v_BandClass_CDMA2000_Cell19,
                                                                                       v_SearchWindowSize_Cell19,
                                                                                       tsc_IdMeasObject_f17,
                                                                                       tsc_IdReportConfig_f17, //@sic R5-123656 sic@
                                                                                       {cs_CellsToAddModCDMA2000(1,v_PNOffset_Cell19), cs_CellsToAddModCDMA2000(2,v_PNOffset_Cell20 )} ) );
    //@siclog "Step 3" siclog@
    //Wait for 30 s to ensure that the UE performs a periodical reporting of HRPD cells
    t_Wait.start ( v_TimerValue_1 );
    // In parallel to steps 3 and 4, the steps specified in table 8.3.2.10.3.2-3 shall take place
    //Parallel behaviour Check: Does the UE transmit a MeasurementReport message to perform periodical intra frequency reporting for HRPD Cell 19?
    t_CheckTimer_1.start( v_TimerValue_2 );
    alt {
      [] t_Wait.timeout{}
      [] t_CheckTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_1Entry(v_PNOffset_Cell19,?) ) )//@sic R5-110734 sic@
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 19, Parallel behaviour to Step 3 and Step 4" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
    }
    //Activate the default to receive periodical reports.
    v_DefaultRef := activate( a_EUTRA_AddDefault_PeriodicalMeasurmentReport( eutra_Cell1, cr_MeasurementReport_CheckAny ) );
    //@siclog "Step 4" siclog@
    //SS sets the cell-specific reference signal levels for Cell 1, Cell 19 and Cell 20 according to row "T1" in table 8.3.2.10.3.2-1.
    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );
    //@siclog "Step 5" siclog@
    //Wait and ignore MeasurementReport messages for 5 s to allow for the switching of cells
    f_Delay( v_TimerValue_2 );
    //Deactivate the default.
    deactivate( v_DefaultRef );
    //@siclog "Step 6" siclog@
    //Wait for 30 s to ensure that the UE performs a periodical reporting of HRPD cells
    t_Wait.start ( v_TimerValue_1 );
    // In parallel to steps 6 and 7, the steps specified in table 8.3.2.10.3.2-4 shall take place
    //Parallel behaviour Check: Does the UE transmit a MeasurementReport message to perform periodical intra frequency reporting for Cell 19 and Cell 17 ?
    t_CheckTimer_1.start( v_TimerValue_2 );
    alt {
      [] t_Wait.timeout{}
      [] t_CheckTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_2Entries(v_PNOffset_Cell19, v_PNOffset_Cell20 ) ) )
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 19, cell 20, Parallel behaviour to Step 6 and Step 7" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
        [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_2Entries(v_PNOffset_Cell20, v_PNOffset_Cell19 ) ) )
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 19, cell 20, Parallel behaviour to Step 6 and Step 7" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
    }
    //Activate the default to receive periodical reports.
    v_DefaultRef := activate( a_EUTRA_AddDefault_PeriodicalMeasurmentReport( eutra_Cell1, cr_MeasurementReport_CheckAny ) );
    //@siclog "Step 7" siclog@
    //The SS changes Cell 1 and Cell 19 parameters according to the row "T2" in table 8.3.2.10.3.2-1.
    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );
    //@siclog "Step 8" siclog@
    //Wait and ignore MeasurementReport messages for 5 s to allow for the switching of cells
    t_Wait.start( v_TimerValue_2 );
    t_Wait.timeout;
    //Deactivate the default.
    deactivate( v_DefaultRef );
    //@siclog "Step 9" siclog@
    //Wait for 30 s to ensure that the UE performs a periodical reporting of HRPD cells
    t_Wait.start ( v_TimerValue_1 );
    // In parallel to steps 9 and 10, the steps specified in table 8.3.2.10.3.2-5 shall take place
    //Parallel behaviour Check: Does the UE transmit a MeasurementReport message to perform periodical intra frequency reporting for Cell 20?
    t_CheckTimer_1.start( v_TimerValue_2 );
    alt {
      [] t_Wait.timeout{}
      [] t_CheckTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_1Entry (v_PNOffset_Cell20,?)) )//@sic R5-110734 sic@
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 20, Parallel behaviour to Step 9 and Step 10" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
    }
    //@siclog "Step 10-11" siclog@
    //SS transmits an RRCConnectionReconfiguration message including measConfig to remove measId for periodic reporting.
    //The UE transmits an RRCConnectionReconfigrationComplete message
    //@sic R5s110743 sic@
   f_EUTRA_RRCConnectionReconfiguration_MeasConfig ( eutra_Cell1,cs_RRCConnectionReconfiguration_MeasRemove( v_RRC_TI ) );
    //Deactivate the default.
    deactivate( v_DefaultRef );
    //@siclog "step 12 siclog@
    //@sic R5-123656 sic@
     t_Wait.start (30.0 ); // 30 seconds wait timer
     alt {
      [] t_Wait.timeout
       {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Periodic measurement reporting stopped in step 12" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CheckAny) ) //
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
    }
    //End TestBody
    //------------------------------------------------
    f_EUTRA_TestBody_Set ( false );
    //tell the CDMA2000 component that the test is over
    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );
    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );
  }; //end of testcase 8.3.2.10
…
After change

…
  //testcase 8.3.2.10:
  // Measurement configuration control and reporting / Inter-RAT measurements / Periodic reporting / Measurement of 1xRTT cells
  function f_TC_8_3_2_10_EUTRA () runs on EUTRA_PTC
  {
    var integer v_T0_T1_T2_RS_EPRE_Cell1 := -60;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var IRAT_CoOrd_SysInfo_Type v_CDMA2000SysInfo;
    var CDMA2000_Type  v_CDMA2000_Type_Cell19;
    var CDMA2000_Type  v_CDMA2000_Type_Cell20;
    var integer v_PNOffset_Cell19;
    var integer v_PNOffset_Cell20;
    var ARFCN_ValueCDMA2000 v_Arfcn_CDMA2000_Cell19;
    var ARFCN_ValueCDMA2000 v_Arfcn_CDMA2000_Cell20;
    var BandclassCDMA2000 v_BandClass_CDMA2000_Cell19;
    var BandclassCDMA2000 v_BandClass_CDMA2000_Cell20;
    var integer v_SearchWindowSize_Cell19 := 15;
    var float v_TimerValue_1;
    var float v_TimerValue_2;
    //@sic R5112614 sic@
    var Frequency_IE_Type v_Frequency_IE_Cell1;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell1 ;
    var default v_DefaultRef := null;
    timer t_Wait;
    timer t_CheckTimer_1;
    //Initialize testcase
    f_EUTRA_Init ( c6 );
    //@sic R5112614 sic@
    v_Frequency_IE_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell1);
    v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_IE_Cell1.DL_ChBandwidth, v_Frequency_IE_Cell1.UL_ChBandwidth);
    v_TimerValue_1 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, nonProtocolTimer, 30.0 );
    v_TimerValue_2 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, nonProtocolTimer, 5.0 );
    //Set EPRE for T0
    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, v_T0_T1_T2_RS_EPRE_Cell1 );
    //Create and configure Cell1
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    f_EUTRA_Preamble ( eutra_Cell1, STATE2_IDLEUPDATE );
    f_EUTRA_RbEst_Def (eutra_Cell1);
    // First Get Cell Parameters for CDMA2000 cell19
    v_CDMA2000SysInfo := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );
    v_CDMA2000_Type_Cell19 := f_ConvertFromBoolToCDMA2000_Type(v_CDMA2000SysInfo.Cdma2000[0].IsHRPD);
    v_PNOffset_Cell19 := v_CDMA2000SysInfo.Cdma2000[0].PNOffset;
    v_Arfcn_CDMA2000_Cell19 := v_CDMA2000SysInfo.Cdma2000[0].Arfcn;
    v_BandClass_CDMA2000_Cell19 := f_ConvertFromIntegerToBandClass(v_CDMA2000SysInfo.Cdma2000[0].BandClass);
    //v_SearchWindowSize_Cell19 := v_CDMA2000SysInfo.Cdma2000[0].SearchWindowSize;
    // Next Get Cell Parameters for CDMA2000 cell20
    v_CDMA2000SysInfo := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );
    v_CDMA2000_Type_Cell20 := f_ConvertFromBoolToCDMA2000_Type(v_CDMA2000SysInfo.Cdma2000[0].IsHRPD);
    v_PNOffset_Cell20 := v_CDMA2000SysInfo.Cdma2000[0].PNOffset;
    v_Arfcn_CDMA2000_Cell20 := v_CDMA2000SysInfo.Cdma2000[0].Arfcn;
    v_BandClass_CDMA2000_Cell20 := f_ConvertFromIntegerToBandClass(v_CDMA2000SysInfo.Cdma2000[0].BandClass);
    //v_SearchWindowSize_Cell20 := v_CDMA2000SysInfo.Cdma2000[0].SearchWindowSize;
    f_EUTRA_TestBody_Set ( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 1-2" siclog@
    //The SS transmits an RRCConnectionReconfiguration message including measConfig to setup measurements and periodical reporting for HRPD cells.
    //The UE transmits an RRCConnectionReconfigurationComplete message to confirm measurement on Cell 19.
    //@sic R5s110743 sic@
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig( eutra_Cell1, cs_RRCConnectionReconfiguration_IratPerMeas_C2K ( v_RRC_TI,
                                                                                       v_Frequency_IE_Cell1.UL_DL_Earfcn.dl_CarrierFreq, //@sic R5112614 sic@
                                                                                       v_ChBandDependency_Cell1.AllowedMeasBandwidth, //@sic R5112614 sic@
                                                                                       v_CDMA2000_Type_Cell19,
                                                                                       v_Arfcn_CDMA2000_Cell19,
                                                                                       v_BandClass_CDMA2000_Cell19,
                                                                                       v_SearchWindowSize_Cell19,
                                                                                       tsc_IdMeasObject_f17,
                                                                                       tsc_IdReportConfig_f17, //@sic R5-123656 sic@
                                                                                       {cs_CellsToAddModCDMA2000(1,v_PNOffset_Cell19), cs_CellsToAddModCDMA2000(2,v_PNOffset_Cell20 )} ) );
    //@siclog "Step 3" siclog@
    //Wait for 30 s to ensure that the UE performs a periodical reporting of HRPD cells
    t_Wait.start ( v_TimerValue_1 );
    // In parallel to steps 3 and 4, the steps specified in table 8.3.2.10.3.2-3 shall take place
    //Parallel behaviour Check: Does the UE transmit a MeasurementReport message to perform periodical intra frequency reporting for HRPD Cell 19?
    t_CheckTimer_1.start( v_TimerValue_2 );
    alt {
      [] t_Wait.timeout
        {
          t_CheckTimer_1.stop;
        }
      [] t_CheckTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_1Entry (v_PNOffset_Cell19,omit) ) )
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 19, Parallel behaviour to Step 3 and Step 4" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
    }
    //Activate the default to receive periodical reports.
    v_DefaultRef := activate( a_EUTRA_AddDefault_PeriodicalMeasurmentReport( eutra_Cell1, cr_MeasurementReport_CheckAny ) );
    //@siclog "Step 4" siclog@
    //SS sets the cell-specific reference signal levels for Cell 1, Cell 19 and Cell 20 according to row "T1" in table 8.3.2.10.3.2-1.
    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );
    //@siclog "Step 5" siclog@
    //Wait and ignore MeasurementReport messages for 5 s to allow for the switching of cells
    f_Delay( v_TimerValue_2 );
    //Deactivate the default.
    deactivate( v_DefaultRef );
    //@siclog "Step 6" siclog@
    //Wait for 30 s to ensure that the UE performs a periodical reporting of HRPD cells
    t_Wait.start ( v_TimerValue_1 );
    // In parallel to steps 6 and 7, the steps specified in table 8.3.2.10.3.2-4 shall take place
    //Parallel behaviour Check: Does the UE transmit a MeasurementReport message to perform periodical intra frequency reporting for Cell 19 and Cell 17 ?
    t_CheckTimer_1.start( v_TimerValue_2 );
    alt {
      [] t_Wait.timeout
        {
          t_CheckTimer_1.stop;
        }
      [] t_CheckTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_1Entry (v_PNOffset_Cell19,omit) ) )
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 19, Parallel behaviour to Step 6 and Step 7" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_1Entry (v_PNOffset_Cell20,omit) ) )
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 20, Parallel behaviour to Step 6 and Step 7" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
    }
    //Activate the default to receive periodical reports.
    v_DefaultRef := activate( a_EUTRA_AddDefault_PeriodicalMeasurmentReport( eutra_Cell1, cr_MeasurementReport_CheckAny ) );
    //@siclog "Step 7" siclog@
    //The SS changes Cell 1 and Cell 19 parameters according to the row "T2" in table 8.3.2.10.3.2-1.
    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );
    //@siclog "Step 8" siclog@
    //Wait and ignore MeasurementReport messages for 5 s to allow for the switching of cells
    t_Wait.start( v_TimerValue_2 );
    t_Wait.timeout;
    //Deactivate the default.
    deactivate( v_DefaultRef );
    //@siclog "Step 9" siclog@
    //Wait for 30 s to ensure that the UE performs a periodical reporting of HRPD cells
    t_Wait.start ( v_TimerValue_1 );
    // In parallel to steps 9 and 10, the steps specified in table 8.3.2.10.3.2-5 shall take place
    //Parallel behaviour Check: Does the UE transmit a MeasurementReport message to perform periodical intra frequency reporting for Cell 20?
    t_CheckTimer_1.start( v_TimerValue_2 );
    alt {
      [] t_Wait.timeout
        {
          t_CheckTimer_1.stop;
        }
      [] t_CheckTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" ); 
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CDMA2000_1Entry (v_PNOffset_Cell20,omit)) )//@sic R5-110734 sic@
        {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Measurement Report for cell 20, Parallel behaviour to Step 9 and Step 10" );
          t_CheckTimer_1.stop;
          t_CheckTimer_1.start( v_TimerValue_2 );
          repeat;
        }
    }
    //@siclog "Step 10-11" siclog@
    //SS transmits an RRCConnectionReconfiguration message including measConfig to remove measId for periodic reporting.
    //The UE transmits an RRCConnectionReconfigrationComplete message
    //@sic R5s110743 sic@
   f_EUTRA_RRCConnectionReconfiguration_MeasConfig ( eutra_Cell1,cs_RRCConnectionReconfiguration_MeasRemove( v_RRC_TI ) );
    //Deactivate the default.
    deactivate( v_DefaultRef );
    //@siclog "step 12 siclog@
    //@sic R5-123656 sic@
     t_Wait.start (30.0 ); // 30 seconds wait timer
     alt {
      [] t_Wait.timeout
       {
          f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Periodic measurement reporting stopped in step 12" );
        }
      [] SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_MeasurementReport_CheckAny) ) //
        {
          f_EUTRA_SetVerdictFailOrInconc( __FILE__, __LINE__, "" );
        }
    }
    //End TestBody
    //------------------------------------------------
    f_EUTRA_TestBody_Set ( false );
    //tell the CDMA2000 component that the test is over
    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );
    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );
  }; //end of testcase 8.3.2.10
…
Change 3 – Correction to function ‘f_TC_8_3_2_10_CDMA2000()’ 
	Function name
	f_TC_8_3_2_10_CDMA2000()

	Reason for change
	As per TS 36.523-1 Table 8.3.2.10.3.2-2, cell 19 has to be switched off at T2 (step 7) to ensure that Test Purpose 2 is fulfilled. However, the TTCN just set the “power Ior” of cell 19 to -90 dBm.

	Summary of change
	Set “power Ior” of cell 19  to “tsc_NonSuitableOffCellPowerIor” instead of “tsc_T2_Cell19”.
Switching OFF is done via a constant imported from “CDMA2000_Constants”.

	TTCN module
	8_3\RRC_Measurement_C2K_CDMA2000.ttcn

	MCC160 Comment
	


Before change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2012, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: LTE_IWD_12wk48
// $Date: 2012-07-02 00:15:45 +0200 (Mon, 02 Jul 2012) $
// $Rev: 6970 $
/******************************************************************************/
module RRC_Measurement_C2K_CDMA2000 {
  import from CommonIratDefs all;
  import from CommonDefs all;
  import from CDMA2000_CommonFunctions all;
  import from CDMA2000_Component all;
  import from CDMA2000_CommonDefs all;
  //****************************************************************
  //Common functions
  //****************************************************************
…
  //================================================================
  // Measurement configuration control and reporting / inter RAT measurements: Periodic reporting (measurement of 1xRTT cells)
  function f_TC_8_3_2_10_CDMA2000 () runs on CDMA2000_PTC
  {
    const PowerIor_Type tsc_T0_T1_Cell19 := -75; // cell power corresponds to Ior
    const PowerIor_Type tsc_T2_Cell19 := -90;
    const PowerIor_Type tsc_T0_Cell20 := -90;
    const PowerIor_Type tsc_T1_T2_Cell20 := -75;
    //Initialise test case
    f_CDMA2000_Init ( EUTRA_CDMA2000, type1XRTT );
    // Set MAX reference power corresponding to Ior/Ioc = 0 for cell 19 and 20 respectively
    f_CDMA2000_CellInfo_InitMaxReferencePower(cdma2000_Cell19, -75, -75, omit); //Ioc = -75, Ior/Ioc = 0, Pilot offset = -7
    f_CDMA2000_CellInfo_InitMaxReferencePower(cdma2000_Cell20, -75, -75, omit); //Ioc = -75, ior/Ioc = 0, Pilot offset = -7
    //Create CDMA2000 cell19 and cell20 and broadcast System Infomation
    f_CDMA2000CreateCell ( cdma2000_Cell19 );
    f_CDMA2000CreateCell ( cdma2000_Cell20 );
    //Set CDMA2000 cell's power level for T0
    f_CDMA2000_SetCellPower( cdma2000_Cell19, tsc_T0_T1_Cell19 );
    f_CDMA2000_SetCellPower( cdma2000_Cell20, tsc_T0_Cell20 );
    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell19 );
    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell20 );
    f_CDMA2000_TestBody_Set( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 4" siclog@
    //First Trigger: Step 4: Change cell power according to T1
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_SetCellPower( cdma2000_Cell20, tsc_T1_T2_Cell20 );
    //@siclog "Step 7" siclog@
    //Second Trigger: Step 7: Change cell power according to T2
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_SetCellPower( cdma2000_Cell19, tsc_T2_Cell19 );
    //Wait for the end of the test
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    //End TestBody
    //------------------------------------------------
    f_CDMA2000_TestBody_Set ( false );
    f_CDMA2000_ReleaseCells(); //Delete all configured CDMA2000 cells
  }; // end of test case 8.3.2.10
…
After change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2012, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: LTE_IWD_12wk48
// $Date: 2012-07-02 00:15:45 +0200 (Mon, 02 Jul 2012) $
// $Rev: 6970 $
/******************************************************************************/
module RRC_Measurement_C2K_CDMA2000 {
  import from CommonIratDefs all;
  import from CommonDefs all;
  import from CDMA2000_CommonFunctions all;
  import from CDMA2000_Component all;
  import from CDMA2000_CommonDefs all;
  import from CDMA2000_Constants all;
  //****************************************************************
  //Common functions
  //****************************************************************
…
  //================================================================
  // Measurement configuration control and reporting / inter RAT measurements: Periodic reporting (measurement of 1xRTT cells)
  function f_TC_8_3_2_10_CDMA2000 () runs on CDMA2000_PTC
  {
    const PowerIor_Type tsc_T0_T1_Cell19 := -75; // cell power corresponds to Ior
    const PowerIor_Type tsc_T2_Cell19 := -90;
    const PowerIor_Type tsc_T0_Cell20 := -90;
    const PowerIor_Type tsc_T1_T2_Cell20 := -75;
    //Initialise test case
    f_CDMA2000_Init ( EUTRA_CDMA2000, type1XRTT );
    // Set MAX reference power corresponding to Ior/Ioc = 0 for cell 19 and 20 respectively
    f_CDMA2000_CellInfo_InitMaxReferencePower(cdma2000_Cell19, -75, -75, omit); //Ioc = -75, Ior/Ioc = 0, Pilot offset = -7
    f_CDMA2000_CellInfo_InitMaxReferencePower(cdma2000_Cell20, -75, -75, omit); //Ioc = -75, ior/Ioc = 0, Pilot offset = -7
    //Create CDMA2000 cell19 and cell20 and broadcast System Infomation
    f_CDMA2000CreateCell ( cdma2000_Cell19 );
    f_CDMA2000CreateCell ( cdma2000_Cell20 );
    //Set CDMA2000 cell's power level for T0
    f_CDMA2000_SetCellPower( cdma2000_Cell19, tsc_T0_T1_Cell19 );
    f_CDMA2000_SetCellPower( cdma2000_Cell20, tsc_T0_Cell20 );
    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell19 );
    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell20 );
    f_CDMA2000_TestBody_Set( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 4" siclog@
    //First Trigger: Step 4: Change cell power according to T1
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_SetCellPower( cdma2000_Cell20, tsc_T1_T2_Cell20 );
    //@siclog "Step 7" siclog@
    //Second Trigger: Step 7: Change cell power according to T2
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_SetCellPower( cdma2000_Cell19, tsc_NonSuitableOffCellPowerIor );
    //Wait for the end of the test
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    //End TestBody
    //------------------------------------------------
    f_CDMA2000_TestBody_Set ( false );
    f_CDMA2000_ReleaseCells(); //Delete all configured CDMA2000 cells
  }; // end of test case 8.3.2.10
…  

5. Execution Log Files

5.1 LG Electronics Optimus G UE 

The LG Electronics Optimus G UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 18 and C2K BC 6. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_8_3_2_10_LG-Electronics_FDD.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
References

	[1]
	R5s120914:    Supporting information for agreement of TC 8.3.2.10. This archive comprises:

                        - text format execution log files
                        - draft Prose CR (“draft_R5-130xxx_TC_8-3-2-10.doc”) .

